Kapitel 17

Integration

Während das Bilden von Ableitungen (mathematisch gesehen) verhältnismäßig einfach ist, stellt der umgekehrte Prozess häufig ein unlösbares Problem dar. Es gibt keine einheitliche Vorgehensweise, die für alle Integrale zum Ziel führt. Zum manuellen Integrieren (ohne Maple) werden normalerweise ganze Bücher mit Tabellen bekannter Integrale verwendet. Wenn das Integral nicht auf schon bekannte Einzelteile zurückgeführt werden kann, werden verschiedene Integrationsregeln angewendet, die in ganz speziellen Fällen zum Ziel führen. Und gar nicht selten kommt es vor, dass sich ein Integral nicht in geschlossener Form darstellen lässt.

Wie sieht nun die Integration in Maple aus? Das Programm kennt eine Reihe von Integrationsalgorithmen, die je nach Typ des Integranden eingesetzt werden. Dazu kommen Tabelleneinträge für elementare Funktionen sowie Platzhalter für Integrale, die prinzipiell nur numerisch berechnet werden können (z.B. die Gaußsche Fehlerfunktion).

Das vorliegende Kapitel beschreibt im Wesentlichen nur ein einziges Kommando, nämlich `int`. Im Verlauf des Kapitels werden verschiedene Anwendungsformen dieses Kommandos vorgestellt, beispielsweise zur numerischen Integration, zur Durchführung von Doppel- und Mehrfachintegralen, zur Integration komplexer Funktionen etc.

`int`
integriert die angegebene Funktion und setzt gegebenenfalls die Integrationsgrenzen ein.

`evalf(Int(..))`
verwendet ein rein numerisches Integrationsverfahren.

`residue`
berechnet Residuen (zur Integration komplexer Funktionen).

Das Integral

Die Integralrechnung ist vielleicht eine der wichtigsten Anwendungen für ein CAS: Statt in endlosen Tabellen nachzuschlagen und sich mit Integrationsregeln herumzuplagen, gibt man einen Befehl ein. Was in der Forschung höchst willkommen ist, hat aber auch Auswirkungen auf die Lehre. Inwieweit müssen Schüler und Studenten von morgen noch die Integration beherrschen? Neben der Fähigkeit eines CAS, Integrale symbolisch zu berechnen, spielt aber – einmal mehr – die Grafik eine wichtige Rolle und man könnte sich die Einführung des Integrals in der Physik etwa so vorstellen:

Im Wechselstromkreis sind Spannung und Strom meistens phasenverschoben. Welche Arbeit wird verrichtet?

Das Produkt aus Spannung und Strom stellt den Momentanwert der Leistung P dar. Für genügend kurze Zeitintervalle Δt kann man die Arbeit durch $\Delta W = P\,\Delta t$ annähern. Das würde dann so aussehen:

```
P:='sin(t)*cos(t-Pi/5)';
```

$$P := \sin(t)\cos(t - \frac{1}{5}\,\pi)$$

```
student[leftbox](P, t=0..5, 30,
                 shading=yellow);
```



Und die Arbeit während einer größeren Zeitspanne müsste dann ungefähr gleich der Summe der einzelnen Beiträge sein (die Summanden werden von Maple umgeformt):

```
W:=student[leftsum](P, t=0..5, 30);
```

$$\mathrm{W} := \frac{1}{6}\left(\sum_{i=0}^{29} \sin(\frac{1}{6}\,i)\sin(\frac{1}{6}\,i + \frac{3}{10}\,\pi)\right)$$

```
evalf(%);
```

$$1.890470645$$

Natürlich fangen dann alle Schüler (aber auch Physikstudenten) sofort an, die Zerlegung zu verfeinern. Einige ändern auch die Grenzen und wundern sich über negative Ergebnisse und die Experimentierfreudigsten ändern die Phasenverschiebung von Spannung und Strom und strahlen, wenn die Arbeit 0 herauskommt. Dann wird es Zeit, vom Einzelfall auf einen allgemein gültigen Zusammenhang zu schließen.

Kann Maple einen Term für die Arbeit finden, wenn man die Zerlegung beliebig fein macht? Man müsste erst die Summe mit veränderlicher oberer Grenze n bilden.

```
Wn:=student[leftsum](P, t=0..5, n);
```

$$Wn := \frac{5}{n}\sum_{i=0}^{n-1} \sin(5\,\frac{i}{n})\sin(5\,\frac{i}{n} + \frac{3}{10}\,\pi)$$

Und dann den Grenzwert für $n \to \infty$ berechnen. Und man bekommt tatsächlich nicht nur einen Term, sondern auch die zugehörige Dezimalzahl, die etwas über der Näherung liegt.

```
limit(Wn,n=infinity); evalf(%);
```

$$\frac{1}{4}\sin(\frac{3}{10}\,\pi) + \frac{5}{2}\cos(\frac{3}{10}\,\pi) - \frac{1}{4}\sin(10 + \frac{3}{10}\,\pi)$$

$$1.921365057$$

Und weil man solche Berechnungen oft benötigt, schreibt man kurz (wobei Maple wieder den Integranden umformt):

```
Int(P, t=0..5) = int(P, t=0..5); evalf(%);
```

$$\int_0^5 \sin(t)\sin(t + \frac{3}{10}\,\pi)\,dt = \frac{1}{4}\sin(\frac{3}{10}\,\pi) + \frac{5}{2}\cos(\frac{3}{10}\,\pi) - \frac{1}{4}\sin(10 + \frac{3}{10}\,\pi)$$

$$1.921365057 = 1.921365057$$

Wenn man keine Integrationsgrenzen angibt, erhält man eine Stammfunktion.

```
Int(P,t) = int(P, t);
```

$$\int \sin(t)\sin(t + 3/10\,\pi)dt = 1/2\,\cos(3/10\,\pi)t - 1/4\,\sin(2\,t + 3/10\,\pi)$$

$$\mathrm{W} := \frac{1}{2}\cos(\frac{3}{10}\,\pi)\,t - \frac{1}{4}\sin(2\,t + \frac{3}{10}\,\pi)$$

Zeichnet man eine Schar solcher Stammfunktionen zusammen mit dem Integranden in ein Schaubild, so liegt die Vermutung nahe, dass der Integrand die Ableitung einer Stammfunktion ist.

```
plot({P,seq(W+n*eval(W,t=0),n=-3..0)},
        t=0..2*Pi);
```




```
'P'=Diff('W',t); P=diff(W,t);
```

$$P = \frac{\partial}{\partial t}\,\mathrm{W}$$

$$\sin(t)\sin(t + \frac{3}{10}\,\pi) = \frac{1}{2}\cos(\frac{3}{10}\,\pi) - \frac{1}{2}\cos(2\,t + \frac{3}{10}\,\pi)$$

```
combine(diff(W,t)-P,trig);
```

$$0$$

Im Worksheet wird noch in Analogie zum Übergang vom Differenzenquotienten zum Differentialquotienten das bestimmte Integral berechnet.

Einfache Anwendungen

Krummlinig begrenzte Flächen

Ein Aufgabentyp, mit dem sich heute noch viele Schüler plagen, ist die Berechnung von Flächen zwischen Kurven. Mit Maple muss man dazu nur über den Betrag der Differenz integrieren. Zum Vergleich: Das Integral über die Differenz ergibt die orientierte Fläche.

```
flaeche:=int(abs(f(x)-g(x)),x=a..b);
```

$$flaeche := \int_a^b |\mathrm{f}(x) - \mathrm{g}(x)|\, dx$$

```
orflaeche:=int(f(x)-g(x),x=a..b);
```

$$orflaeche := \int_a^b \mathrm{f}(x) - \mathrm{g}(x)\, dx$$

Nun trifft Maple die Fallunterscheidungen, wie ein einfaches Beispiel demonstriert.

```
f:=x->x: g:=x->x/2+1: flaeche; orflaeche;
```

$$\left(\begin{cases} b - \frac{1}{4}\,b^2 & b \le 2 \\ \frac{1}{4}\,b^2 - b + 2 & 2 < b \end{cases}\right) - \left(\begin{cases} a - \frac{1}{4}\,a^2 & a \le 2 \\ \frac{1}{4}\,a^2 - a + 2 & 2 < a \end{cases}\right)$$

$$\frac{1}{4}\,b^2 - \frac{1}{4}\,a^2 - b + a$$

Ein etwas anspruchsvolleres Beispiel wird im Worksheet ausführlich dargestellt. Die Fläche kann dabei mit zwei Befehlen berechnet werden. Für die Schattierung muss man allerdings ein wenig in die Trickkiste greifen.

```
f:=x->-x^3+4*x; g:=x->x^4-4*x^2+1;
```

$$f := x \to -x^3 + 4\,x$$

$$g := x \to x^4 - 4\,x^2 + 1$$



Rotationskörper

Auch Rotationskörper sind ein beliebtes Prüfungsthema, wobei das Problem eigentlich nie in der Anwendung der Formel liegt, sondern eher in der räumlichen Vorstellung. Ein gefundenes Fressen also für ein CAS.

Eine Fläche soll um die x-Achse rotieren. Man kann den Schwierigkeitsgrad noch etwas erhöhen, wenn die Fläche von zwei Kurven begrenzt wird (hier zwei Wurzelfunktionen). Die folgenden Befehle dienen zunächst nur der grafischen Darstellung:

```
kurven:=plot({sqrt(x),sqrt(2*(x-1))},x=0..3,thickness=2,color=black):
schnitt2d:=polygonplot([[0,0],seq([x,sqrt(2*(x-1))],x=seq(1+i/20,i=0..20)),
seq([x,sqrt(x)],x=seq(2-i/20,i=0..40))],axes=normal,tickmarks=[2,2],color=cyan):
display(kurven,schnitt2d);
```

Mit ähnlichen Befehlen lässt sich auch die 3D-Grafik erstellen, sodass man sich nun recht gut vorstellen kann, was zu berechnen ist.





Und wie lautet nun das Ergebnis?

```
RotVol:=(radius,a,b)->Pi*int(radius(x)^2,x=a..b);
```

$$RotVol := (radius,\, a,\, b) \to \pi \int_a^b radius(x)^2\, dx$$

```
Aussen:=RotVol(x->sqrt(x),0,2);
```

$$Aussen := 2\,\pi$$

```
Innen:=RotVol(x->sqrt(2*(x-1)),1,2);
```

$$Innen := \pi$$

```
Aussen-Innen;
```

$$\pi$$

Uneigentliche Integrale

Es existieren zwei Typen uneigentlicher Integrale: solche, bei denen die Integrationsgrenzen unendlich sind, und solche, bei denen im Integrationsbereich der Funktionswert Unendlich auftritt. Integrale des ersten Typs werden häufig anstandslos mit `int` berechnet. Allerdings kann es sein, dass bei der Integration über ein unbeschränktes Intervall zusätzliche Annahmen erforderlich sind:

```
int(exp(a*x),x=0..infinity);
```

```
Definite integration: Can't determine if the integral is convergent.
Need to know the sign of --> -a
Will now try indefinite integration and then take limits.
```

$$\lim_{x \to \infty} \frac{e^{(a\,x)} - 1}{a}$$

Mit `assume(a>0)` oder `assume(a<0)` kann man hier Maple weiterhelfen.

Integriert man über Singularitäten, so kann es sein, dass das Integral nicht definiert ist. Im ersten Beispiel konvergiert das Integral. Im zweiten existiert der Cauchysche Hauptwert.

```
int(1/sqrt(abs(x)),x=-1..1);
```

$$4$$

```
int(1/x^3, x=-3..2);
```

$$\int_{-3}^{2} \frac{1}{x^3}\,dx$$

```
int(1/x^3, x=-3..2,
    CauchyPrincipalValue);
```

$$\frac{-5}{72}$$

Maple kommt auch mit dem Fall zurecht, dass im Integrationsbereich mehrere Singularitäten mit Cauchyschem Hauptwert liegen, wie im nebenstehenden Beispiel an den Punkten $x = -1$ und $x = 0$.

```
int(1/(x^2+x), x=-2..1,
    CauchyPrincipalValue);
```

$$-2\,\ln(2)$$

```
evalf(%);
```

$$-1.386294361$$

Die Abbildung zeigt den Verlauf der Funktion.

```
plot( 1/(x^2+x), x=-2..1, -20..20);
```



Hinweis: Im Worksheet finden Sie ein Beispiel für die Integration über Unstetigkeitsstellen mit endlichem Sprung.

Integraltabellen und Integrationsregeln

Tabellen

Nach dem Motto 'jedem seine Tabelle' kann man sich eine Sammlung seiner Lieblingsintegrale zusammenstellen. Auch wenn man jedes Integral neu berechnen kann, ist das oft zu Vergleichszwecken sinnvoll, besonders wenn es sich nicht um elementare Funktionen handelt, für die man oft zusätzliche Annahmen machen muss. Ein Spreadsheet leistet hier gute Dienste, weil die Berechnung automatisiert werden kann. Einfach in der ersten Spalte eine neue oder leicht geänderte Funktion eintragen und das Spreadsheet neu auswerten. Einen zusätzlichen Komfort stellt die Programmierbarkeit der Spreadsheets durch Befehle im Worksheet dar.

	A	B	C
1	$\mathrm{f}(x)$	$\int \mathrm{f}(x)\,dx$	$\int\int \mathrm{f}(x)\,dx\,dx$
2	x^n	$\frac{x^{(n+1)}}{n+1}$	$\frac{x^{(n+2)}}{(n+1)\,(n+2)}$
3	$e^{(-x^2)}$	$\frac{1}{2}\sqrt{\pi}\,\mathrm{erf}(x)$	$\frac{1}{2}\sqrt{\pi}\,(x\,\mathrm{erf}(x)+\frac{e^{(-x^2)}}{\sqrt{\pi}})$
4	$\frac{1}{\sqrt{1-x^2}\,\sqrt{1-k^2\,x^2}}$	$\mathrm{EllipticF}(x,\,\mathrm{csgn}(k)\,k)$	*ein etwas umfangreicher Term*
5	$\mathrm{Dirac}(x)\,\mathrm{f}(x)$	$\mathrm{f}(0)\,\mathrm{Heaviside}(x)$	$\mathrm{f}(0)\,\mathrm{Heaviside}(x)\,x$

Integrationsregeln

Die Integrationsregeln sind seit Maple eigentlich kein Thema mehr, deshalb stehen sie auch im Package für Studenten. Dennoch soll es ja vorkommen, dass auch Maple eine Stammfunktion, die es eigentlich geben müsste, nicht findet. Dann kann man sich immerhin bei einer Substitution oder einer partiellen Integration beraten lassen:

```
with(student):
changevar(x=sin(u), Int(sqrt(1-x^2), x=a...b), u);
```

$$\int_{\arcsin(a)}^{\arcsin(b)} \sqrt{1-\sin(u)^2}\cos(u)\,du$$

```
changevar(x=g(u), Int(f(x), x=a...b), u);
```

$$\int_{\mathrm{RootOf}(\mathrm{g}(\_Z)-a)}^{\mathrm{RootOf}(\mathrm{g}(\_Z)-b)} \mathrm{f}(\mathrm{g}(u))\,(\frac{\partial}{\partial u}\,\mathrm{g}(u))\,du$$

```
intparts(Int(u(x)*v(x),x),u(x));
```

$$\mathrm{u}(x)\int \mathrm{v}(x)\,dx - \int (\frac{\partial}{\partial x}\,\mathrm{u}(x))\int \mathrm{v}(x)\,dx\,dx$$

```
int(diff(f(x),x)/f(x),x);
```

$$\ln(\mathrm{f}(x))$$

Kurvenintegrale

Kurvenintegrale können z.B. zur Berechnung krummlinig begrenzter Flächen oder zur Berechnung von Bogenlängen verwendet werden. Das Standardbeispiel ist die Ellipse.

Flächen

Man kann zunächst eine allgemein gültige Formel für die Berechnung von Flächen in parametrisierter Form aufstellen (es gibt drei Formen davon). Anschließend legt man den Weg fest und wertet das Integral aus.

```
kurvint:=-int(y(t)*diff(x(t),t),t=ta..tb);
```

$$\mathit{kurvint} := -\int_{ta}^{tb} \mathrm{y}(t)\,(\frac{\partial}{\partial t}\,\mathrm{x}(t))\,dt$$

```
x:=t->a*cos(t)+3; y:=t->b*sin(t)+2; kurvint; Flaeche:=eval(kurvint,{ta=0,tb=2*Pi});
```

$$x := t \to a\cos(t)+3$$

$$y := t \to b\sin(t)+2$$

$$-\frac{1}{2}\,a\,b\cos(tb)\sin(tb)+\frac{1}{2}\,a\,b\,tb-2\,a\cos(tb)+\frac{1}{2}\,a\,b\cos(ta)\sin(ta)-\frac{1}{2}\,a\,b\,ta+2\,a\cos(ta)$$

$$\mathit{Flaeche} := a\,b\,\pi$$

Im vorliegenden Beispiel beginnt und endet der Integrationsweg im Punkt $(5|2)$ und verläuft im Gegenuhrzeigersinn. In der ersten Hälfte wird dabei die gesamte Fläche der Figur berechnet und in der zweiten Hälfte die dunklere Fläche abgezogen.



Bogenlängen

Auch hier zunächst die allgemeine Formel für die Bogenlänge in parametrisierter Form.

```
bogen:=int(sqrt(diff(x(t),t)^2+diff(y(t),t)^2),t=0..2*Pi);
```

$$bogen := \int_0^{2\pi} \sqrt{(\frac{\partial}{\partial t}\,\mathrm{x}(t))^2 + (\frac{\partial}{\partial t}\,\mathrm{y}(t))^2}\,dt$$

```
x:=t->a*cos(t)+3; y:=t->b*sin(t)+2; bogen;
```

$$x := t \to a\cos(t) + 3$$

$$y := t \to b\sin(t) + 2$$

$$4\,\frac{\mathrm{EllipticE}(\sqrt{-\frac{-a^2+b^2}{a^2}})\,\sqrt{\frac{b^2}{a^2}}\,a^2}{\sqrt{b^2}}$$

Leider scheint diesmal das Ergebnis nicht so einfach zu sein wie bei der Berechnung der Ellipsenfläche: Es ist ein elliptisches Integral, das sich nicht geschlossen angeben lässt.

```
assume(b>0,a>0): simplify(bogen); eval(bogen,{a=2,b=1}); evalf(%);
```

$$4\,\mathrm{EllipticE}(\frac{\sqrt{a^{\sim 2} - b^{\sim 2}}}{a^{\sim}})\,a^{\sim}$$

$$8\,\mathrm{EllipticE}(\frac{1}{2}\sqrt{3})$$

$$9.688448216$$

Dass die Berechnung solcher Integrale auch ihre Tücken haben kann, wird im Worksheet gezeigt: Wenn man das Integral mit variablen Grenzen ansetzt, müssen wegen der Periodizität von `EllipticE` die passenden Annahmen gemacht werden (was wiederum von der Reihenfolge der Auswertung abhängt).

Integration komplexer Funktionen, Residuen

Analytische Funktionen können wie Funktionen einer reellen Veränderlichen integriert werden, da in diesem Fall der Wert des Integrals vom Weg unabhängig ist.

```
int(z^2,z=1..I);
```

$$\frac{-1}{3} - \frac{1}{3} I$$

```
int( z*cos(z^2), z=1..I);
```

$$-\sin(1)$$

Im allgemeinen Fall kann der Integrationsweg parametrisiert angegeben werden.

```
kint:=int(f(z(t))*diff(z(t),t),t=ta..tb);
```

$$kint := \int_{ta}^{tb} \mathrm{f}(\mathrm{z}(t)) \left(\frac{\partial}{\partial t} \mathrm{z}(t)\right) dt$$

So kann man z.B. den Einheitskreis um den Ursprung als Integrationsweg wählen.

```
z:=t->cos(t)+I*sin(t); kint;
```

$$z := t \to \cos(t) + I \sin(t)$$

$$\int_{ta}^{tb} \mathrm{f}(\cos(t) + I\sin(t)) \, (-\sin(t) + I\cos(t)) \, dt$$

Und zum Vergleich mit dem ersten Beispiel die Funktion z^2 integrieren. (Weitere Beispiele finden Sie im Worksheet.)

```
f:=z->z^2; eval(kint,{ta=0,tb=Pi/2});
```

$$f := z \to z^2$$

$$\frac{-1}{3} - \frac{1}{3} I$$

Der Absolutbetrag von z ist nicht analytisch. Das erste Ergebnis erhält man, wenn man längs der geradlinigen Verbindung von $(1|0)$ nach $(0|I)$ integriert, das zweite $(-1 + I)$, wenn man einen Viertelkreis wählt.

$$f := \mathrm{abs}$$

$$-\frac{1}{4}\sqrt{2}\ln(1+\sqrt{2}) - \frac{1}{2} + \frac{1}{2} I + \frac{1}{4} I \sqrt{2} \ln(1+\sqrt{2})$$

$$-1 + I$$

Nicht sehr hilfreich ist hier der Vorschlag von Maple, wenn man ohne die Angabe eines Wegs integriert.

```
z:='z': int(abs(z),z=1..I);
```

$$\lim_{z \to I-} \left(\begin{cases} -\frac{1}{2} z^2 & z \le 0 \\ \frac{1}{2} z^2 & 0 < z \end{cases} \right) - \frac{1}{2}$$

Hinweis: Eine Funktion kann mit `linalg[laplacian]` auf Analytizität untersucht werden.

Residuen

Wegen der Wegunabhängigkeit des Integrals analytischer Funktionen verschwinden Umlaufintegrale über solche Funktionen, falls der Weg keine Singularitäten umschließt. In diesem Fall hat man über die Residuen zu summieren (Residuensatz):

$$\oint_C f(z)\,dz = 2\pi I \sum_{j=1}^{k} \operatorname*{Res}_{z=z_j} f(z)$$

Natürlich kann man mit Maple das Umlaufintegral auch explizit berechnen. Aber schon beim Standardbeispiel, der Funktion $1/z$, erhält man 0 statt $2\pi I$, wenn man obige Definition des Wegintegrals (`kint`) verwendet. Der Grund dafür liegt darin, dass Maple mit dieser Formulierung zuerst die logarithmische Integration anwendet.

Wenn man aber die logarithmische Integration vor Maple versteckt, so wird das Umlaufintegral um die Singularität von $1/z$ korrekt berechnet. Ein weiteres Beispiel also, dass es bei einem CAS oft auf die Reihenfolge der Anwendung von Regeln ankommt.

```
z:=cos(t) + I*sin(t): dz:=diff(z,t);
```

$$dz := -\sin(t) + I\cos(t)$$

```
f:=1/z;
```

$$f := \frac{1}{\cos(t) + I\sin(t)}$$

```
int( f*dz, t=0..2*Pi);
```

$$2\,I\,\pi$$

Natürlich wird man bei Umlaufintegralen um Singularitäten den Residuensatz verwenden. Dennoch lohnt es sich, an dieser Stelle die Integration im Komplexen etwas näher zu beleuchten.

Die nebenstehende Funktion f hat drei Singularitäten. (Mit dem Befehl `discont` erhält man nur reelle Unstetigkeitsstellen.)

```
f:=(2*z^3+3*z+50)/(z^3-3*z-52);
```

$$f := \frac{2\,z^3 + 3\,z + 50}{z^3 - 3\,z - 52}$$

```
solve(denom(f));
```

$$4,\ -2 + 3\,I,\ -2 - 3\,I$$

Die Residuen kann man an der Reihenentwicklung ablesen oder direkt mit `residue` bestimmen.

```
series(f,z=4,3);
```

$$\frac{38}{9}\,(z-4)^{-1} + \frac{29}{27} + \mathrm{O}(z-4)$$

```
residue(f,z=4);
```

$$\frac{38}{9}$$

Im Worksheet finden Sie Befehle, mit denen man die Funktion (im Bild der Realteil) und den Integrationsweg darstellen kann. Außerdem werden verschiedene Integrationswege (und die Rechengeschwindigkeit) getestet.



Mehrfachintegrale

Weil die Welt nicht eindimensional ist, sind Mehrfachintegrale das tägliche Brot des Ingenieurs und Physikers: Die Gebiete und Räume, in denen Mehrfachintegrale eingesetzt werden, sind so umfassend, dass sie in dieser kurzen Einführung nur angedeutet werden können.

Wir beginnen mit einem einfachen Beispiel, der Berechnung des Volumens einer Pyramide. Die Aufgabe besteht immer darin, ein passendes Volumenelement zu finden, die Integrationsgrenzen richtig zu wählen und in der richtigen Reihenfolge zu integrieren.



Im `student`-Package gibt es eine Kurzfassung für das Dreifachintegral. Es ist zweckmäßig, solche Rechnungen zuerst an einfachen Beispielen zu testen, die man mit einer bekannten Formel überprüfen kann. Außerdem lässt sich die Integration auch zerlegen (siehe Worksheet).

```
with(student):
Tripleint(1,z=0..2-2*x-2*y,y=0..1-x,
                              x=0..1);
```

$$\int_0^1 \int_0^{1-x} \int_0^{2-2\,x-2\,y} 1\,dz\,dy\,dx$$

```
value(%);
```

$$\frac{1}{3}$$

Während man das Volumen einer Pyramide auch in der Formelsammlung nachschlagen kann, wird es beim Trägheitsmoment schon etwas komplizierter:

```
Tripleint(y^2+z^2,z=0..2-2*x-2*y,y=0..1-x,x=0..1);
```

$$\int_0^1 \int_0^{1-x} \int_0^{2-2\,x-2\,y} y^2 + z^2\,dz\,dy\,dx$$

```
map(value,%);
```

$$\int_0^1 -\frac{7}{6}(1-x)^4+\frac{1}{3}(10-10\,x)(1-x)^3+\frac{1}{2}(\frac{1}{3}(2-2\,x)(-8+8\,x)-\frac{2}{3}(2-2\,x)^2)(1-x)^2$$
$$+\frac{1}{3}(2-2\,x)^3(1-x)dx$$

```
value(%);
```

$$\frac{1}{6}$$

Allgemein werden in Maple also Dreifachintegrale folgendermaßen formuliert:

```
Int(Int(Int(g(x,y,z),z=z1(x,y)..z2(x,y)),y=y1(x)..y2(x)),x=x1..x2);
```

$$\int_{x1}^{x2}\int_{\mathrm{y1}(x)}^{\mathrm{y2}(x)}\int_{\mathrm{z1}(x,\,y)}^{\mathrm{z2}(x,\,y)} \mathrm{g}(x,\,y,\,z)\,dz\,dy\,dx$$

Dabei sind kartesische Koordinaten in der Praxis eher die Ausnahme. Meistens müssen für die Integration die Koordinaten dem Problem angepasst werden. Dazu benötigt man die Funktionaldeterminante der Koordinatentransformation. In Maple kann mit ’`linalg[jacobian]`’ die erforderliche Matrix aufgestellt werden:

```
restart: with(linalg): xyz:=[x(u,v,w),y(u,v,w),z(u,v,w)];
```

$$xyz := [\mathrm{x}(u,\,v,\,w),\,\mathrm{y}(u,\,v,\,w),\,\mathrm{z}(u,\,v,\,w)]$$

```
jacobian(xyz, [u,v,w]);
```

$$\begin{bmatrix} \frac{\partial}{\partial u}\mathrm{x}(u,\,v,\,w) & \frac{\partial}{\partial v}\mathrm{x}(u,\,v,\,w) & \frac{\partial}{\partial w}\mathrm{x}(u,\,v,\,w) \\ \frac{\partial}{\partial u}\mathrm{y}(u,\,v,\,w) & \frac{\partial}{\partial v}\mathrm{y}(u,\,v,\,w) & \frac{\partial}{\partial w}\mathrm{y}(u,\,v,\,w) \\ \frac{\partial}{\partial u}\mathrm{z}(u,\,v,\,w) & \frac{\partial}{\partial v}\mathrm{z}(u,\,v,\,w) & \frac{\partial}{\partial w}\mathrm{z}(u,\,v,\,w) \end{bmatrix}$$

```
fundet:=det(jacobian(xyz, [u,v,w])):
```

Ein bekanntes Beispiel (Kugelkoordinaten):

```
x:=(u,v,w)-> u*cos(v)*sin(w);
y:=(u,v,w)-> u*sin(v)*sin(w);
z:=(u,v,w)-> u*cos(w);
```

$$x := (u,\,v,\,w) \to u\cos(v)\sin(w)$$

$$y := (u,\,v,\,w) \to u\sin(v)\sin(w)$$

$$z := (u,\,v,\,w) \to u\cos(w)$$

```
fundet;
```

$$-\cos(v)^2 \sin(w)^3 u^2 - \sin(v)^2 \sin(w)^3 u^2 - \cos(w)^2 u^2 \sin(v)^2 \sin(w)$$
$$- \cos(w)^2 u^2 \cos(v)^2 \sin(w)$$

```
simplify(%);
```

$$-\sin(w)\, u^2$$

Numerische Integration

Durch das Kommando `evalf` kann eine numerische Auswertung des Integrals erreicht werden. Dabei ist zwischen zwei Varianten zu unterscheiden: Bei `evalf(int(..))` wird das Integral zuerst (soweit möglich) symbolisch berechnet, erst das Ergebnis wird numerisch ausgewertet. Zusammen mit der trägen Variante des Integrationskommandos, also in der Form `evalf(Int(..))`, verwendet Maple ein rein numerisches Verfahren zur Integration. In den meisten Fällen kommen beide Varianten zum gleichen Ergebnis (eventuell mit kleinen Rundungsfehlern), es kann aber auch der Fall eintreten, dass die eine oder andere Variante zum falschen Ergebnis führt. Insofern stellen die beiden Berechnungsarten – sofern beide durchführbar sind – eine recht brauchbare Kontrolle des Ergebnisses dar.

In diesem Zusammenhang ist noch eine Anmerkung zum Unterschied zwischen `int` und `Int` erforderlich: Die Kommandos `Int`, `Diff`, `Sum` und `Product` heißen träge Kommandos, weil sie eine Auswertung des mathematischen Kommandos blockieren. Sie entfalten ihre Wirkung erst in Zusammenarbeit mit anderen Kommandos, etwa mit `evalf` zur numerischen Auswertung des ganzen Ausdrucks. Siehe auch Kapitel 28, Abschnitt 'Der Aufbau von Maple'.

Häufig stellt die numerische Integration die einzige Möglichkeit dar, zu einem Ergebnis zu gelangen.

```
int( sin(1/x^3), x=1..2);
```

$$\int_1^2 \sin(x^{-3})dx$$

```
evalf(%);
```

$$0.3548334332$$

Maple verwendet zur numerischen Integration je nach Art der mathematischen Funktion eines von drei Integrationsverfahren, das durch die Angabe einer Option im vierten Parameter bestimmt werden kann: Zur Auswahl stehen `_CCquad` (Clenshaw-Curtis Quadraturverfahren, Defaulteinstellung), `_Dexp` (doppelt-exponentielles Verfahren) oder `_NCrule` (Newton-Cote-Verfahren). Normalerweise ist ein Abweichen von dem Verfahren, das Maple auswählt, nicht sinnvoll – in der Regel ergeben sich dadurch nur (zum Teil deutlich) höhere Rechenzeiten.

```
evalf(Int(sin(x)/x, x=1..100, 15, _NCrule));
```

$$0.616142396521873$$

Wenn die symbolische Integration nicht durchgeführt wird, kann das auch an `RootOfs` liegen:

```
f:=1/(1+x^7);
```

$$f := \frac{1}{1+x^7}$$

```
int( f, x=1..2);
```

$$\frac{1}{7}\ln(3) - \frac{1}{7}\left(\sum_{\_R=\%1} \_R\ln(2-\_R)\right) - \frac{1}{7}\ln(2) + \frac{1}{7}\left(\sum_{\_R=\%1} \_R\ln(1-\_R)\right)$$

$$\%1 := \mathrm{RootOf}(\_Z^6 - \_Z^5 + \_Z^4 - \_Z^3 + \_Z^2 - \_Z + 1)$$

```
evalf(%);
```

$$.1163017046 + 0.\,I$$

Auch hier empfiehlt sich also die rein numerische Integration:

```
evalf( Int(f, x=1..2), 20);
```

$$.11630170437709733349$$

Wenn man uneigentliche Integrale numerisch berechnet, kann man mit `evalf(Limit( ))` zu erstaunlichen Ergebnissen kommen:

```
int( 1/x^2, x=0..infinity);
```

$$\infty$$

```
evalf( Int( 1/x^2, x=0..infinity) );
```

$$\mathrm{Float}(\infty)$$

```
evalf( Limit(Int( 1/x^2, x=n..infinity), n=0) );
```

$$2.000000000$$

Kontrolle der Integration

Der wirkungsvollste Kontrollmechanismus besteht darin, das Integral auf verschiedene Weisen zu berechnen. Bei komplexen Integralen bietet sich etwa eine alternative Berechnung über den Residuensatz an. Bei bestimmten Integralen ist häufig eine rein numerische Integration möglich, die intern nach einem vollkommen eigenständigen Algorithmus durchgeführt wird (achten Sie auf die Großschreibung von `Int`!).

Des Weiteren kann es bei bestimmten Integralen zweckmäßig sein, das dazugehörige allgemeine Integral zu berechnen und (getrennt für Real- und Imaginärteil) mit `plot` zu zeichnen. Wenn dabei ein Sprung auftritt, ist das bestimmte Integral mit allergrößter Wahrscheinlichkeit falsch. Jetzt kann eventuell versucht werden, die Unstetigkeitsstelle durch links- und rechtsseitige Grenzwerte zu eliminieren.

Die einfachste Kontrolle allgemeiner Integrale besteht darin, diese wieder zu differenzieren und mit der Ausgangsfunktion zu vergleichen. Das Bilden der Ableitung ist ein mathematisch verhältnismäßig einfacher Vorgang, sodass zumindest dabei vorausgesetzt werden kann, dass Maple fehlerfrei arbeitet. Aber auch diese scheinbar elegante Lösung kann ihre Tücken haben. Erstens können sich Unstetigkeitsstellen im Integral durch das Differenzieren wieder herausheben. Zweitens kann die Rechenzeit bei komplizierten Integralen zu einem Hindernis werden. Und drittens kann es passieren, dass die Ausgangsfunktion und die Ableitung des Integrals vollkommen unterschiedlich aussehen, mathematisch aber übereinstimmen. Dann kann es oft einige Zeit dauern, bis man die richtigen Befehle zur Termumformung findet, besonders wenn trigonometrische Funktionen im Spiel sind:

```
f:=sin(a+1/x);
```

$$f := \sin(a + \frac{1}{x})$$

```
inte:=int(f,x);
```

$$inte := \sin(a + \frac{1}{x})\, x - \mathrm{Si}(-\frac{1}{x}) \sin(a) - \mathrm{Ci}(\frac{1}{x}) \cos(a)$$

```
integrand:=diff(inte,x);
```

$$integrand := -\frac{\cos(a + \frac{1}{x})}{x} + \sin(a + \frac{1}{x}) - \frac{\sin(\frac{1}{x}) \sin(a)}{x} + \frac{\cos(\frac{1}{x}) \cos(a)}{x}$$

```
simplify(integrand);
```

$$\frac{-\cos(\frac{a\,x+1}{x}) + \sin(\frac{a\,x+1}{x})\, x - \sin(\frac{1}{x}) \sin(a) + \cos(\frac{1}{x}) \cos(a)}{x}$$

```
expand(integrand);
```

$$\sin(a) \cos(\frac{1}{x}) + \cos(a) \sin(\frac{1}{x})$$

```
combine(%,trig);
```

$$\sin(a + \frac{1}{x})$$

Wenn eine Vereinfachung nicht gelingt, können Sie die Identität zweier Ausdrücke durch das Einsetzen von Zufallszahlen testen. Dieses Verfahren ist natürlich nicht ganz zuverlässig. Die Wahrscheinlichkeit, dass zwei Ausdrücke mathematisch gleichwertig sind, wenn sie für eine Abfolge von zehn Zufallszahlen übereinstimmen, ist aber immerhin sehr hoch. Wichtig ist natürlich, dass die Zufallszahlen dem Problem entsprechen, d.h. im allgemeinsten Fall komplexe Fließkommazahlen aus allen vier Quadranten der komplexen Zahlenebene. Manchmal kann es auch sinnvoll sein, ganze Zahlen oder Brüche einzusetzen und dann eine symbolische Vereinfachung durchzuführen.

Das Beispiel unten stellt eine Fortsetzung des obigen Beispiels dar. In *textexpr* werden für x und a mehrere Fließkomma-Zufallszahlen im Bereich ± 5000 eingesetzt. Die beiden abschließenden Kommandos berechnen die größte absolute Abweichung von 0. Diese Vorgehensweise wäre vor allem bei umfangreicheren Testreihen sinnvoll.

```
testexpr:=integrand-f;
```

$$testexpr := -\frac{\cos(a+\frac{1}{x})}{x} - \frac{\sin(\frac{1}{x})\sin(a)}{x} + \frac{\cos(\frac{1}{x})\cos(a)}{x}$$

```
test:='5000-rand()/1e8';
```

$$test := 5000 - .1\,10^{-7}\,\mathrm{rand}()$$

```
seq( evalf(subs(x=test, a=test, testexpr)), i=1..5);
```

$$-.2893\,10^{-9},\ .24\,10^{-11},\ .6581\,10^{-9},\ .450\,10^{-10},\ .2134\,10^{-9}$$

```
map(abs, [%] );
```

$$[.2893\,10^{-9},\ .24\,10^{-11},\ .6581\,10^{-9},\ .450\,10^{-10},\ .2134\,10^{-9}]$$

```
max( op(%) );
```

$$.6581\,10^{-9}$$

Maple beim Integrieren zusehen

Wenn Sie wissen möchten, wie Maple zu seinen Ergebnissen kommt, können Sie beim Integrieren zusehen. Dazu müssen Sie die Systemvariable `infolevel` für `int` auf einen Wert zwischen 2 oder 5 stellen (je höher der Wert, desto ausführlicher werden die Informationen). Das erste Beispiel zeigt Maple bei der Integration von $\frac{1}{1+\ln(x)}$.

```
infolevel[int]:=3:
int( 1/(1+ln(x)), x);
```

```
int/indef1:   first-stage indefinite integration
int/indef2:   second-stage indefinite integration
int/ln:   case of integrand containing ln
int/indef1:   first-stage indefinite integration
int/indef2:   second-stage indefinite integration
int/exp:   case of integrand containing exp
```

$$-e^{(-1)}\,\mathrm{Ei}(1,\ -\ln(x)-1)$$

Im zweiten Beispiel wird die Funktion $\frac{sin(x)}{x}$ im Bereich zwischen 1 und 2 numerisch integriert. In diesem Fall muss `infolevel` für das Kommando `‘evalf/int‘` erhöht werden.

```
infolevel[‘evalf/int‘]:=2:
evalf(Int(sin(x)/x, x=1..2));
```

```
evalf/int/control:   integrating on 1 .. 2 the integrand
```

$$\frac{\sin(x)}{x}$$

```
evalf/int/control:   Trying easyproc
evalf/int/control:   from ccquad: error = .9167111514330e-12
integrand evals = 19
result = .6593299064355
```

$$.6593299064$$

Syntaxzusammenfassung

`int(f,x);` `int(f,x=a..b);` `int(f,x=a..b,opt);`
berechnet das unbestimmte oder bestimmte Integral der Funktion f. Wenn über Singularitäten integriert werden soll, kann die Option `CauchyPrincipalValue` verwendet werden – `int` liefert dann den Cauchyschen Hauptwert.

`Doubleint(g,x=a..b,y =c..d);` `Tripleint(g,x=a..b,y=c..d,z=e..f)`
sind Befehle aus dem Package `student` und berechnen das Doppel- bzw. Dreifachintegral von f (mit und ohne Grenzen). Die träge Integration wird mit `value` ausgewertet.

`evalf(Int(f,x=a..b));`
führt eine rein numerische Integration durch.

`residue(f,x=x0);`
berechnet das Residuum von f an der Stelle $x0$.

`with(linalg):`
`det(jacobian([x,y,z], [u,v,w]));`
bildet die Funktionaldeterminante der Funktionen x, y, z bezüglich u, v, w (wird bei Koordinatentransformationen benötigt).

`laplacian(f, [x,y,z], coords=c);`
kann dazu verwendet werden, die Funktion f auf Analytizität zu testen (mit optionaler Angabe des Koordinatensystems).